

# FORM & Compton scattering. (November 16, 2019)

## A. Context

Figure (1) gives the Feynman diagrams corresponding to the two lowest-order processes contributing to Compton scattering. As per Griffiths' (2008, Example 7.4) the probability amplitude  $\mathcal{M}_1$  associated with the first of those Feynman diagrams is:

$$\frac{\mathcal{M}_1}{i} = \int \frac{d^4q}{(2\pi)^4} \underbrace{\epsilon_{\mu(2)}}_{\gamma \text{ in}} \left[ \underbrace{\bar{u}_4}_{e^- \text{ out}} \underbrace{ig_e \gamma^\mu}_{\text{vertex}} \underbrace{\frac{i(\not{q} + mc)}{q^2 - m^2c^2}}_{\text{propogator}} \underbrace{ig_e \gamma^\kappa}_{\text{vertex}} \underbrace{u_1}_{e^- \text{ in}} \right] \underbrace{\epsilon_{\kappa(3)}^*}_{\gamma \text{ out}} \times (2\pi)^8 \delta^4(p_1 - p_3 - q) \delta^4(p_2 + q - p_4), \quad (1)$$

where  $\bar{u}_4$  is the (adjoint) spinor associated with the outgoing electron,  $u_1$  is the spinor representing the incoming electron,  $\epsilon_{\mu(2)}$  are the components of the polarization 4-vector for the incoming photon (etc.), and the 'slash' notation  $\not{q} \equiv q^\lambda \gamma_\lambda$ . Spin and momentum labels have been dropped, i.e. more completely, for example,  $\bar{u}_4 = \bar{u}^{(s_4)}(p_4)$  with two possible spin states ( $s_4 = 1, 2$ ). (Implicitly, the  $mc$  factor is multiplied by the identity matrix.)




Figure 1: Feynman diagram of the two lowest-order (i.e. 2-vertex) processes contributing to Compton scattering (Griffiths 2008, p246). Associated with the diagram on the left is an amplitude  $\mathcal{M}_1$ .

Integration sends  $q \rightarrow (p_1 - p_3)$ , and we drop the factor  $(2\pi)^4 \delta^4(p_1 + p_2 - p_3 - p_4)$ . Then noting that for fixed index  $\mu$  the quantity  $\epsilon_{\mu(2)}$  is just a scalar (and ditto  $\epsilon_{\kappa(3)}^*$ ) so that  $\epsilon_{\mu(2)} \bar{u}_4 = \bar{u}_4 \epsilon_{\mu(2)}$  (etc.), the amplitude associated with diagram 1 can be written

$$\mathcal{M}_1 = \frac{g_e^2}{(p_1 - p_3)^2 - m^2c^2} [\bar{u}_4 \not{\epsilon}_2 (\not{p}_1 - \not{p}_3 + mc) u_1 \not{\epsilon}_3^*], \quad (2)$$

where  $\not{\epsilon}_2 \equiv \epsilon_{\mu(2)}\gamma^\mu$ . Analogous steps for the second diagram give amplitude

$$\mathcal{M}_2 = \frac{g_e^2}{(p_1 + p_2)^2 - m^2c^2} [\bar{u}_4 \not{\epsilon}_3^* (\not{p}_1 + \not{p}_2 + mc) u_1 \not{\epsilon}_2] . \quad (3)$$

To evaluate the cross section for Compton scattering the quantity

$$|\mathcal{M}|^2 = \mathcal{M}_1\mathcal{M}_1^* + \mathcal{M}_2\mathcal{M}_2^* + \mathcal{M}_1\mathcal{M}_2^* + \mathcal{M}_2\mathcal{M}_1^* \quad (4)$$

is needed. After some manipulation one obtains

$$\mathcal{M}_1\mathcal{M}_1^* = A^2 \bar{u}_4 \not{\epsilon}_2 [\not{p}_1 - \not{p}_3 + mc] \not{\epsilon}_3^* u_1 \bar{u}_1 \not{\epsilon}_3 [\not{p}_1 - \not{p}_3 + mc] \not{\epsilon}_2^* u_4 , \quad (5)$$

$$\mathcal{M}_2\mathcal{M}_2^* = B^2 \bar{u}_4 \not{\epsilon}_3^* [\not{p}_1 + \not{p}_2 + mc] \not{\epsilon}_2 u_1 \bar{u}_1 \not{\epsilon}_2^* [\not{p}_1 + \not{p}_2 + mc] \not{\epsilon}_3 u_4 , \quad (6)$$

$$\mathcal{M}_1\mathcal{M}_2^* = AB \bar{u}_4 \not{\epsilon}_2 [\not{p}_1 - \not{p}_3 + mc] \not{\epsilon}_3^* u_1 \bar{u}_1 \not{\epsilon}_2^* [\not{p}_1 + \not{p}_2 + mc] \not{\epsilon}_3 u_4 , \quad (7)$$

$$\mathcal{M}_2\mathcal{M}_1^* = AB \bar{u}_4 \not{\epsilon}_3^* [\not{p}_1 + \not{p}_2 + mc] \not{\epsilon}_2 u_1 \bar{u}_1 \not{\epsilon}_3 [\not{p}_1 - \not{p}_3 + mc] \not{\epsilon}_2^* u_4 , \quad (8)$$

where

$$A = \frac{g_e^2}{(p_1 - p_3)^2 - m^2c^2} = \frac{g_e^2}{-2p_1 \cdot p_3} , \quad (9)$$

$$B = \frac{g_e^2}{(p_1 + p_2)^2 - m^2c^2} = \frac{g_e^2}{2p_1 \cdot p_2} \quad (10)$$

(because  $p_1^2 = m^2c^2$ ,  $p_2^2 = p_3^2 = 0$ ).

Then averaging over the initial spin states and summing over final spin states, one obtains the four contributions

$$\langle |\mathcal{M}_1|^2 \rangle = \frac{A^2}{4} Q_{\mu\lambda} Q_{\nu\kappa} \text{Tr} [\gamma^\mu (\not{p}_1 - \not{p}_3 + mc) \gamma^\nu (\not{p}_1 + mc) \gamma^\kappa (\not{p}_1 - \not{p}_3 + mc) \gamma^\lambda (\not{p}_4 + mc)] , \quad (11)$$

$$\langle |\mathcal{M}_2|^2 \rangle = \frac{B^2}{4} Q_{\mu\lambda} Q_{\nu\kappa} \text{Tr} [\gamma^\mu (\not{p}_1 + \not{p}_2 + mc) \gamma^\nu (\not{p}_1 + mc) \gamma^\kappa (\not{p}_1 + \not{p}_2 + mc) \gamma^\lambda (\not{p}_4 + mc)] , \quad (12)$$

$$\langle \mathcal{M}_1\mathcal{M}_2^* \rangle = \frac{AB}{4} Q_{\mu\lambda} Q_{\nu\kappa} \text{Tr} [\gamma^\mu (\not{p}_1 - \not{p}_3 + mc) \gamma^\nu (\not{p}_1 + mc) \gamma^\lambda (\not{p}_1 + \not{p}_2 + mc) \gamma^\kappa (\not{p}_4 + mc)] , \quad (13)$$

$$\langle \mathcal{M}_2\mathcal{M}_1^* \rangle = \frac{AB}{4} Q_{\mu\lambda} Q_{\nu\kappa} \text{Tr} [\gamma^\mu (\not{p}_1 + \not{p}_2 + mc) \gamma^\nu (\not{p}_1 + mc) \gamma^\lambda (\not{p}_1 - \not{p}_3 + mc) \gamma^\kappa (\not{p}_4 + mc)] , \quad (14)$$

to

$$\langle \mathcal{M}^2 \rangle = \langle |\mathcal{M}_1|^2 \rangle + \langle |\mathcal{M}_2|^2 \rangle + \langle \mathcal{M}_1\mathcal{M}_2^* \rangle + \langle \mathcal{M}_2\mathcal{M}_1^* \rangle . \quad (15)$$

In Eqs. (11-14) the tensor  $Q_{\mu\lambda}$  is defined

$$Q_{\mu\lambda} = \sum_{s=1,2} \epsilon_\mu^{(s)} \epsilon_\lambda^{(s)*} ,$$

where the photon polarization vector  $\epsilon^\mu = (\epsilon^0, \epsilon)$  with  $\epsilon^0 = 0$ . The completeness relation (for the photon polarization three vectors) is

$$\sum_{s=1,2} \epsilon_i^{(s)} \epsilon_j^{(s)*} = \delta_{ij} - \hat{p}_i \hat{p}_j \quad (16)$$

where  $\hat{p}_i$  is the unit vector  $\hat{p}_i = p_i/|p_i|$  (no summation). Hence

$$Q_{\mu\lambda} = \begin{cases} 0, & \text{if } \mu = 0 \text{ or } \lambda = 0 \\ \delta_{\mu\lambda} - \hat{p}_\mu \hat{p}_\lambda, & \text{otherwise.} \end{cases} \quad (17)$$

Eqs. (11–17) concur with the solution manual written by D. Griffiths (available at [www.academia.edu](http://www.academia.edu) and elsewhere).

### A. Evaluation of $\langle |\mathcal{M}|^2 \rangle$ in terms of Mandelstam variables, using FORM

It can be seen from Eqs. (11, 12) that  $\langle |\mathcal{M}_1|^2 \rangle$  and  $\langle |\mathcal{M}_2|^2 \rangle$  are equivalent under the substitution  $p_2 \leftrightarrow -p_3$  (or  $s \leftrightarrow u$  in terms of the Mandelstam variables  $s \equiv (p_1 + p_2)^2$  and  $u \equiv (p_2 - p_4)^2$  used below), and the same symmetry links  $\langle \mathcal{M}_1 \mathcal{M}_2^* \rangle$  and  $\langle \mathcal{M}_2 \mathcal{M}_1^* \rangle$ .

\*\*\* compton\_scattering.frm JDW 29 Oct. 2019 \*\*\*

Vectors p1, p2, p3, p4;

Symbols m,s,u,[ge^4],[2ge^4],denu,dens;

Indices alpha,beta,gamma,delta,rho,kappa,lambda,mu,nu;

Off Statistics;

```
Local traceM1M1 = d_(mu,lambda)*d_(nu,kappa)
                  *g_(1,mu)*(g_(1,p1)-g_(1,p3)+m)
                  *g_(1,nu)*(g_(1,p1)+m)
                  *g_(1,kappa)*(g_(1,p1)-g_(1,p3)+m)
                  *g_(1,lambda)*(g_(1,p4)+m);
```

```
Local traceM2M2 = d_(mu,lambda)*d_(nu,kappa)
                  *g_(2,mu)*(g_(2,p1)+g_(2,p2)+m)
                  *g_(2,nu)*(g_(2,p1)+m)
                  *g_(2,kappa)*(g_(2,p1)+g_(2,p2)+m)
                  *g_(2,lambda)*(g_(2,p4)+m);
```

```
Local traceM1M2str = d_(mu,lambda)*d_(nu,kappa)
                    *g_(3,mu)*(g_(3,p1)-g_(3,p3)+m)
```

```

    *g_(3,nu)*(g_(3,p1)+m)
    *g_(3,lambda)*(g_(3,p1)+g_(3,p2)+m)
    *g_(3,kappa)*(g_(3,p4)+m);

Local traceM2M1str = d_(mu,lambda)*d_(nu,kappa)
    *g_(4,mu)*(g_(4,p1)+g_(4,p2)+m)
    *g_(4,nu)*(g_(4,p1)+m)
    *g_(4,lambda)*(g_(4,p1)-g_(4,p3)+m)
    *g_(4,kappa)*(g_(4,p4)+m);

Local M1M1 =    div_(1,2)*[2ge^4]*traceM1M1*div_(1,4)*(-1/(2*p1.p3))^2;

Local M2M2 =    div_(1,2)*[2ge^4]*traceM2M2*div_(1,4)*( 1/(2*p1.p2))^2;

Local M1M2str = div_(1,2)*[2ge^4]*traceM1M2str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local M2M1str = div_(1,2)*[2ge^4]*traceM2M1str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

Local [<|M|^2>]=M1M1+M2M2+M1M2str+M2M1str;

Trace4,1;
Trace4,2;
Trace4,3;
Trace4,4;
.sort;

repeat;
id p2.p2=0;
id p3.p3=0;
id p1.p1=m^2;
id p4.p4=m^2;
endrepeat;

Print;
.sort;

*** Introduce Mandelstam variables.
*** id s= 2*p1.p2+m^2;

```

```

*** id u=-2*p1.p3+m^2;
***

repeat;
id 1/(p1.p2)=2/(s-m^2);
id 1/(p1.p3)=-2/(u-m^2);
id p1.p4 = (u+s)/2;
id p2.p3=(u+s)/2-m^2;
id p2.p4 = (m^2-u)/2;
id p3.p4 = (s-m^2)/2;
endrepeat;
.sort;

repeat;
id s/(s-m^2)=1+m^2/(s-m^2);
id 1/(s - m^2)/(s + m^2)=1/(s^2-m^4);
endrepeat;
.sort;

id 1/(u-m^2)=denu;
id 1/(s-m^2)=dens;
.sort;

id 1/(s - m^2)/(u - m^2)*u^2=dens*denu*u^2;
.sort;

repeat;
id denu*u = 1+m^2*denu;
endrepeat;
.sort;

Bracket [2ge^4],m;

Print;
.sort;
.end;

```

Here are some selected outputs from the program.

$$\begin{aligned}
\text{traceM1M1} = & \\
& + m^2 * ( - 16*u - 80*p1.p3 ) \\
& + m^4 * ( 48 ) \\
& + 16*p1.p3*s;
\end{aligned}$$

$$\begin{aligned}
\text{traceM2M2} = & \\
& + m^2 * ( - 16*s + 80*p1.p2 ) \\
& + m^4 * ( 48 ) \\
& - 16*p1.p2*u;
\end{aligned}$$

showing the expected symmetry of  $|\mathcal{M}_1|^2$  and  $|\mathcal{M}_2|^2$ . The result for  $\langle |\mathcal{M}|^2 \rangle$  is (correctly)

$$\begin{aligned}
[\langle |\mathcal{M}|^2 \rangle] = & \\
& + [2g_e^4] * ( - u*\text{dens} - s*\text{denu} ) \\
& + m^2*[2g_e^4] * ( 5*\text{dens} + 5*\text{denu} ) \\
& + m^4*[2g_e^4] * ( 4*\text{dens}^2 + 8*\text{denu}*\text{dens} + 4*\text{denu}^2 );
\end{aligned}$$

where  $\text{denu} \equiv 1/(u - m^2)$  and  $\text{dens} \equiv 1/(s - m^2)$ . (Thanks to Jos Vermaseren for helpful suggestions.)

## B. Evaluation of the differential cross-section in the Lab frame, using FORM

Defining the lab frame as that in which the electron is initially at rest and taking  $(\omega, \omega')$  as respectively the frequencies of the (incoming, outgoing) photons, the classical (deterministic) formula for Compton scattering is

$$\frac{1}{\omega'} - \frac{1}{\omega} = \frac{\hbar}{mc^2} (1 - \cos \theta) , \quad (18)$$

where the scattering angle  $\theta$  is the angle between the outgoing and incoming photon momenta. This can be easily rearranged to read (with  $\hbar = c = 1$ )

$$- \left( \frac{\omega'}{\omega} \right)^2 \sin^2 \theta = \frac{m^2}{\omega^2} \left( 1 - 2\frac{\omega'}{\omega} + \frac{\omega'^2}{\omega^2} \right) + \frac{2m}{\omega} \left( \frac{\omega'^2}{\omega^2} - \frac{\omega'}{\omega} \right) , \quad (19)$$

an expression that will be used below to simplify the output of a FORM program intended to derive the Klein-Nishina formula for the differential cross section, viz.

$$\frac{d\sigma}{d \cos \theta} = \frac{g_e^4}{16\pi m^2} \left[ \frac{\omega'^3}{\omega^3} + \frac{\omega'}{\omega} - \frac{\omega'^2}{\omega^2} \sin^2 \theta \right] . \quad (20)$$

In the lab. frame the four-momenta of the four particles are:

$$p_1 = (m, 0, 0, 0), \quad (21)$$

$$p_2 = (\omega, 0, 0, \omega) \quad (22)$$

$$p_3 = (\omega', \omega' \sin \theta, 0, \omega' \cos \theta) \quad (23)$$

$$p_4 = (E_{4L}, p_{4x}, p_{4y}, p_{4z}), \quad (24)$$

and it is straightforward to show that

$$s = m^2 + 2m\omega, \quad (25)$$

$$u = m^2 - 2m\omega', \quad (26)$$

$$p_1 \cdot p_2 = p_3 \cdot p_4 = m\omega, \quad (27)$$

$$p_1 \cdot p_3 = p_2 \cdot p_4 = m\omega', \quad (28)$$

$$p_2 \cdot p_3 = \omega\omega' (1 - \cos \theta). \quad (29)$$

The differential cross section in the lab. frame is<sup>1</sup> (Griffiths' (1987, Eq. B.11)

$$\frac{d\sigma}{d\Omega} = \left( \frac{\hbar E_3}{8\pi m c E_2} \right)^2 \langle |\mathcal{M}|^2 \rangle \rightarrow \left( \frac{1}{8\pi m} \right)^2 \left( \frac{\omega'}{\omega} \right)^2 \langle |\mathcal{M}|^2 \rangle. \quad (30)$$

Writing  $d\Omega = -d(\cos \theta) d\phi$  and assuming azimuthal symmetry for the integration  $0 \leq \phi \leq 2\pi$  we obtain

$$\frac{d\sigma}{d \cos \theta} = -2\pi \frac{d\sigma}{d\Omega}. \quad (31)$$

Below, FORM is used to evaluate Eq. (31), checking against Eq. (20).

\*\*\* compton\_scattering\_crossection\_lab.frm JDW 16 Nov. 2019 \*\*\*

\*\*\* Aim: reproduce the Klein-Nishina formula \*\*\*

Vectors p1, p2, p3, p4;

Symbols [2Pi], [2ge^4], [(hbar/8mPi)^2], [ge^4/16Pim^2], [(op/o)^2 sin^2 theta];

Symbols m, omega, omegaprime, o0Nop, op0No, denomOmega, mop0No2, m2op0No3, m2overo2, m2op20No2;

Indices alpha, beta, gamma, delta, rho, kappa, lambda, mu, nu;

Off Statistics;

```
Local traceM1M1 = d_(mu,lambda)*d_(nu,kappa)
                  *g_(1,mu)*(g_(1,p1)-g_(1,p3)+m)
```

---

<sup>1</sup>This is presumably given, too, in the 2nd edition – which however is currently inaccessible to the writer.

```

*g_(1,nu)*(g_(1,p1)+m)
*g_(1,kappa)*(g_(1,p1)-g_(1,p3)+m)
*g_(1,lambda)*(g_(1,p4)+m);

```

```

Local traceM2M2 = d_(mu,lambda)*d_(nu,kappa)
*g_(2,mu)*(g_(2,p1)+g_(2,p2)+m)
*g_(2,nu)*(g_(2,p1)+m)
*g_(2,kappa)*(g_(2,p1)+g_(2,p2)+m)
*g_(2,lambda)*(g_(2,p4)+m);

```

```

Local traceM1M2str = d_(mu,lambda)*d_(nu,kappa)
*g_(3,mu)*(g_(3,p1)-g_(3,p3)+m)
*g_(3,nu)*(g_(3,p1)+m)
*g_(3,lambda)*(g_(3,p1)+g_(3,p2)+m)
*g_(3,kappa)*(g_(3,p4)+m);

```

```

Local traceM2M1str = d_(mu,lambda)*d_(nu,kappa)
*g_(4,mu)*(g_(4,p1)+g_(4,p2)+m)
*g_(4,nu)*(g_(4,p1)+m)
*g_(4,lambda)*(g_(4,p1)-g_(4,p3)+m)
*g_(4,kappa)*(g_(4,p4)+m);

```

```

Local M1M1 = div_(1,2)*[2ge^4]*traceM1M1*div_(1,4)*(-1/(2*p1.p3))^2;

```

```

Local M2M2 = div_(1,2)*[2ge^4]*traceM2M2*div_(1,4)*( 1/(2*p1.p2))^2;

```

```

Local M1M2str = div_(1,2)*[2ge^4]*traceM1M2str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

```

```

Local M2M1str = div_(1,2)*[2ge^4]*traceM2M1str*div_(1,4)*(-1/(2*p1.p3))*(1/(2*p1.p2));

```

```

Local [<|M|^2>]=M1M1+M2M2+M1M2str+M2M1str;

```

```

Local dsigmatadomegaLAB = [(hbar/8mPi)^2]*[<|M|^2>]*omegaprime^2*1/omega^2;

```

```

Local dsigmatadcosthetaLAB = [2Pi]*dsigmatadomegaLAB;

```

```

Trace4,1;

```



```

Trace4,2;
Trace4,3;
Trace4,4;
.sort;

repeat;
id p2.p2=0;
id p3.p3=0;
id p1.p1=m^2;
id p4.p4=m^2;
endrepeat;
.sort;

*** No need to introduce Mandelstam variables, can evaluate the dot products
*** directly in terms of omega, omega' ***

repeat;
id 1/(p1.p2) = 1/(m*omega);
id 1/(p1.p3) = 1/(m*omegaprime);
id p1.p4 = m^2 + m*(omega-omegaprime);
id p2.p3 = m*(omega-omegaprime);
id p2.p4= m*omegaprime;
id p3.p4 = m*omega;
endrepeat;
.sort;

repeat;
*id omega/omegaprime=o0Nop;
*id omegaprime/omega=op0No;
endrepeat;

Print;
.sort;

id 1/omega=denom0Omega;
id denom0Omega*omegaprime=op0No;
.sort;

```

```
id [2Pi]*[2ge^4]*[(hbar/8mPi)^2]=[ge^4/16Pim^2];
.sort;
```

```
Bracket [ge^4/16Pim^2],m,omega;
Print;
.sort;
```

\*\*\* at this point the correct answer is recognisable  
 \*\*\* need now to simplify, pulling in the  $\sin^2(\theta)$  term

```
id denomOmega*op0No*m=mop0No2;
id denomOmega*mop0No2*m=m2op0No3;
id denomOmega*denomOmega*m*m=m2overo2;
id mop0No2*mop0No2=m2op20No2;
.sort;
```

\*\*\* The following four steps amount to a "fudge", to recast the solution, which is recogni  
 \*\*\* into the standard Klein-Nishina expression involving  $\sin^2(\theta)$

```
if(coefficient<0) discard;
if (coefficient > 1) Discard;
id m2op20No2=0;
id m2overo2=-[(op/o)^2 sin^2 theta];
.sort;
```

```
Bracket [ge^4/16Pim^2],m,omega;
```

```
Print;
.sort;
.end;
```

An intermediate output is

```
dsigmadcosthetaLAB =
  + [ge^4/16Pim^2] * ( op0No + op0No^3 )
  + [ge^4/16Pim^2]*m * ( - 2*op0No*denomOmega + 2*op0No^2*denomOmega )
  + [ge^4/16Pim^2]*m^2 * ( denomOmega^2 - 2*op0No*denomOmega^2 + op0No^2*
```

$\text{denom}\omega^2$  );

where for instance  $\omega'$  represents  $\omega'/\omega$ . The expected term in  $\sin^2\theta$  is recognizable, under disguise in the second two lines. How to make the needed transformation? After an operation that amounts (one could say) to the electronic equivalent of using the old typewriter white-out, and that presumably can be achieved elegantly by the experienced user of FORM), one obtains

$\text{dsigmadcosthetaLAB} = [g^4/16\pi^2] * ( \omega' + \omega'^3 - [(\omega'/\omega)^2 \sin^2 \theta] );$

This concurs with the Klein-Nishina formula.

## References

Griffiths, D. 1987. *Introduction to Elementary Particles*. Wiley.

Griffiths, D. 2008. *Introduction to Elementary Particles*. Second edn. Wiley-VCH.